

# Parameterization (Managed Instance)

Last updated by | Vitor Tomaz | Aug 5, 2020 at 12:43 PM PDT

---

```
//*****
// Parameterization
//*****
//PARAM.01
// any parameterization problems
// from XTS
MonWiQueryParamData
| where LogicalServerName =~ '{LogicalServerName}'
| summarize count(), dcount(statement_sql_hash), dcount(query_hash), dcount(query_plan_hash),
avg(compile_cpu_time), max(compile_cpu_time), avg(compile_duration), max(compile_duration) by
logical_database_name, query_param_type
| extend statement_query_ratio = todouble(dcount_statement_sql_hash) / todouble(dcount_query_hash)
| extend parameterization_type = iff(query_param_type == 0, 'NONE', iff(query_param_type == 1, 'USER',
iff(query_param_type == 2, 'SIMPLE', 'FORCED')))
| extend problem_level = iff(query_param_type <> 0 and query_param_type <> 1, 'X', iff(statement_query_ratio
> 1000, 'Extremely High', iff(statement_query_ratio > 100, 'High', iff(statement_query_ratio > 20, 'Medium',
iff(statement_query_ratio > 10, 'Low', 'No problem')))))
| extend compiles_count = count_
| project logical_database_name, query_param_type, parameterization_type, problem_level,
statement_query_ratio, compiles_count, dcount_statement_sql_hash, dcount_query_hash,
dcount_query_plan_hash, avg_compile_cpu_time, max_compile_cpu_time, avg_compile_duration,
max_compile_duration
| order by logical_database_name, query_param_type asc nulls last

// PARAM.02
// individual query hash parameterization types
// from xts
MonWiQueryParamData
| where LogicalServerName =~ '{LogicalServerName}'
| summarize count(), dcount(statement_sql_hash), dcount(template_hash), dcount(query_plan_hash),
avg(compile_cpu_time), max(compile_cpu_time), avg(compile_duration), max(compile_duration),
makeset(query_plan_hash), argmax(originalEventTimestamp, ClusterName), argmax(originalEventTimestamp,
NodeName), argmax(originalEventTimestamp, AppName) by logical_database_name, query_hash,
query_param_type
| extend parameterization_type = iff(query_param_type == 0, 'NONE', iff(query_param_type == 1, 'USER',
iff(query_param_type == 2, 'SIMPLE', 'FORCED')))
| extend compiles_count = count_
| project logical_database_name, query_hash, parameterization_type, query_param_type, compiles_count,
dcount_statement_sql_hash, dcount_template_hash, dcount_query_plan_hash, avg_compile_cpu_time,
max_compile_cpu_time, avg_compile_duration, max_compile_duration, set_query_plan_hash,
last_known_cluster=max_originalEventTimestamp_ClusterName,
last_known_node_name=max_originalEventTimestamp_NodeName,
last_known_app_name=max_originalEventTimestamp_AppName
```

```
| order by compiles_count desc  
| limit 100
```

**How good have you found this content?**



-