

Provision duration of Azure SSIS IR

Last updated by | Wanli Mao | Nov 1, 2022 at 11:56 PM PDT

- SSIS IR don't have public SLA promise for Provision (duration) time.
- SSIS IR have a feature, standby pool, that is used to accelerate Provision speed for customers. However, this only works for IR **not joining Customer's VNet** or IR **joining Customer's VNet via Express VNet injection**. If customer hit standby pool on Provision, the duration may reduce from around 30 ~ 60 minutes to about 5 minutes.

Expectation

- There is no guarantee that customer would always hit the standby pool. It's, however, expected that customer would hit standby pool most of time.
- In below cases, you can reach to PG for investigation
 - Customer's IR was hitting standby pools but suddenly doesn't hit standby pool any more
 - Customer's IR was hitting standby pools in a low rate

Hit Rate

Please fill in correct subscription id, data factory name and IR name. By default, it queries hit rate for past 30 days.

```
let _isExpressVNetInjectionIR = 0; // 0 or 1; 1 means express VNet IR
let _startTime = ago(30d);
let _subscriptionId = '';
let _dataFactoryName = '';
let _irName = '';
AisManagementTaskTelemetryEvent
| where env_time > _startTime
| where subscriptionId == _subscriptionId
| where dataFactoryName == _dataFactoryName and irName == _irName
| where taskType == 'Provision'
| where vNetType == 'Managed' or isVNetV2 == 1
| where isVNetV2 == _isExpressVNetInjectionIR
| summarize count(), hit = countif(isStandbyPoolUsed) by isVNetV2
| extend hitRate = todecimal(hit) / count_
```

Triage

Acceptable hit rate

1. If total provision count < 10, one standby pool hit miss should be regarded as acceptable.
2. If total provision count >= 10, hit rate >= 0.9 should be regarded as acceptable.

Investigation for low hit rate

Please paste query and result from below queries (also query for hit rate) in a thread.

Standby pool size is set per vm size for each pool type in each region.

1. Get pool type, vmSize, region by task id

```
let _taskId = "095d3806-ff73-4189-bc25-380de7e5452eProvision0543124332";
let poolType = AisManagementTaskTraceEvent
| where taskId == _taskId
| project env_time, message
| where message has 'Pool Type is'
| extend PoolType = extract(@"Pool Type is ([a-zA-Z]+)", 1, message)
| project PoolType;
let vmInfo =
AisManagementTaskTelemetryEvent
| where taskId == _taskId
| project vmSize, isVNetV2, env_cloud_location;
union vmInfo, poolType
| project region = env_cloud_location, vmSize, PoolType, isVNetV2
```

2. Please fill in correct value get from last query to check

```

let _poolType = '';
let _region = '';
let _vmSize = '';
AisPoolManagementTaskEvent
| where env_time > ago(20m)
| where env_cloud_location == _region
| project-reorder env_time, message
| where eventType == "GetExperienceValue"
| project env_time, message, env_cloud_location, poolType
| where poolType == _poolType
| where message has 'The VM size'
| extend poolNumber = toint(extract("pool number is ([0-9]+)", 1, message)), vmSize = extract("The VM size is ([A-Za-z_0-9]+)", 1, message)
| distinct env_cloud_location, poolNumber, vmSize
| where vmSize == _vmSize

```

If there is no record or pool number is zero, please ask SSIS PG to increase pool number to 1. Otherwise provide the query and result by following query.

```

let _isExpressVNetInjectionIR = 0; // 0 or 1; 1 means express VNet IR
let _region = '';
let _vmSize = '';
AisManagementTaskTelemetryEvent
| where env_cloud_location == _region
| where vmSize == _vmSize
| where env_time > ago(30d)
| where vNetType == 'Managed' or isVNetV2 == 1
| where isVNetV2 == _isExpressVNetInjectionIR
| where subscriptionId != "4897dd0a-3b36-4503-a76b-fc22b677833c"
| summarize ProvisionCount = count() by subscriptionId

```